// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/interfaces/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (interfaces/IERC20.sol)

pragma solidity ^0.8.0;

import "../token/ERC20/IERC20.sol";

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/ConspiracyDAO.sol



// ********************************************* //
// ********       Our Social Handles     ****** //
// ******************************************* //




// https://twitter.com/TokenConspiracy

// https://t.me/ConspiracyDAO

// Conspiracydaotoken.com




// ********************************************* //


// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/interfaces/IERC20.sol";

contract ConspiracyDAO is Ownable {
    struct Conspiracy{
        uint256 id;
        address creator;
        string name;
        string description;
        string imageUri;
        uint256 upvotes;
        uint256 downvotes;
    }

    address public constant cdaoTokenAddress = 0xADF2c192Ba98A6700E15d412F26e904843363867;
    IERC20 private constant cdaoToken = IERC20(cdaoTokenAddress);

    uint256 public createConspiracyCost;

    Conspiracy[] private conspiracies;

    // voterAddress => conspiracy Id => upvoted bool
    mapping(address => mapping(uint256 => bool)) private upvoted;
    // voterAddress => conspiracy Id => downvoted bool
    mapping(address => mapping(uint256 => bool)) private downvoted;

    //////////////////////////////////////////
    //////////     Events     ///////////////
    ////////////////////////////////////////
    event CreatedConspiracy(uint256 id, address creator, string name, string description, string imageUri);
    event Upvoted(address upvoter, uint256 conspiracyId, uint256 latestUpvotes, uint256 latestDownvotes);
    event Downvoted(address downvoter, uint256 conspiracyId, uint256 latestUpvotes, uint256 latestDownvotes);


    constructor(uint256 _createConspiracyCost) {
        createConspiracyCost = _createConspiracyCost * 1e18;
    }

    //////////////////////////////////////////
    //////////     Modifiers     ////////////
    ////////////////////////////////////////

    modifier shouldNotBeUpvoted(uint256 conspiracyId){
        require(!upvoted[msg.sender][conspiracyId], "You have already Updvoted this conspiracy");
        _;
    }

    modifier shouldNotBeDownvoted(uint256 conspiracyId) {
        require(!downvoted[msg.sender][conspiracyId], "You have already Downvoted this conspiracy");
        _;
    }

    ///////////////////////////////////////////////////
    //////////     External Functions     ////////////
    /////////////////////////////////////////////////

    /**
     * 
     * @param name Conspiracy Name
     * @param description Conspiracy Description
     * @param imageUri Conspiracy Image URI
     * 
     * @notice It checks and deducts the createConspiracyCost from the creators wallet
     */
    function createConspiracy(string calldata name, string calldata description, string calldata imageUri) external {
        // Basic Validations
        require(bytes(name).length > 0, "Name cannot be empty");
        require(bytes(description).length > 0, "Description cannot be empty");
        require(bytes(imageUri).length > 0, "Image URI cannot be empty");
        

        address creator = msg.sender;

        require(cdaoToken.balanceOf(creator) >= createConspiracyCost, "Your CDAO Token Balance is too low to create a conspiracy!");

        uint256 conspiracyId = conspiracies.length;

        Conspiracy memory conspiracy = Conspiracy({
            name: name,
            id: conspiracyId,
            creator: creator,
            description: description,
            imageUri: imageUri,
            upvotes: 0,
            downvotes: 0
        });

        conspiracies.push(conspiracy);

        cdaoToken.transferFrom(creator, owner(), createConspiracyCost);
        
        emit CreatedConspiracy(conspiracyId, creator, name, description, imageUri);
    }

    /**
     * 
     * @param conspiracyId Id of the conspiracy that the user wants to Upvote
     * 
     * @notice This function check if the user has already upvoted and reverts &
     * If the user has downvoted the conspiracy before then it changes it to a upvote
     */
    function upvote(uint256 conspiracyId) external shouldNotBeUpvoted(conspiracyId) {
        address user = msg.sender;
        Conspiracy storage conspiracy = conspiracies[conspiracyId];

        if(downvoted[user][conspiracyId]){
            downvoted[user][conspiracyId] = false;
            conspiracy.downvotes--;
        }
        
        upvoted[user][conspiracyId] = true;
        conspiracy.upvotes++;

        emit Upvoted(user, conspiracyId, conspiracy.upvotes, conspiracy.downvotes);
    }

    /**
     * 
     * @param conspiracyId Id of the conspiracy that the user wants to Downvote
     * 
     * @notice This function check if the user has already Downvoted and reverts &
     * If the user has Upvoted the conspiracy before then it changes it to a Downvote
     */
    function downvote(uint256 conspiracyId) external shouldNotBeDownvoted(conspiracyId){
        address user = msg.sender;
        Conspiracy storage conspiracy = conspiracies[conspiracyId]; 

        if(upvoted[user][conspiracyId]){
            upvoted[user][conspiracyId] = false;
            conspiracy.upvotes--;
        }
        
        downvoted[user][conspiracyId] = true;
        conspiracy.downvotes++;

        emit Downvoted(user, conspiracyId, conspiracy.upvotes, conspiracy.downvotes);
    }

    /**
     * 
     * @param newCreateConspiracyCost New cost for creating conspiracy
     * 
     * @notice Only owner can call this function
     * @notice Multiplies by 1e18 "So enter a normal number and not a token number (without adding 18 extra zeros)"
     */
    function setCreateConspiracyCost(uint256 newCreateConspiracyCost) external onlyOwner {
        createConspiracyCost = newCreateConspiracyCost * 1e18;
    }


    ///////////////////////////////////////////////////
    //////////     External View Functions     ////////////
    /////////////////////////////////////////////////

    function getAllConspiracies() external view returns(Conspiracy[] memory){
        return conspiracies;
    }

    function getConspiracy(uint256 conspiracyId) external view returns(Conspiracy memory){
        return conspiracies[conspiracyId];
    } 

    function getUsersVoteFor(uint256 conspiracyId) external view returns(string memory) {
        address user = msg.sender;
        if(upvoted[user][conspiracyId]){
            return "Upvoted";
        } else if(downvoted[user][conspiracyId]){
            return "Downvoted";
        } else {
            return "Not voted";
        }
    }
}